
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 11.2 which is part of the ‘IMS_34229_IWD_13wk08’ test suite in ATS ‘iwd-EUTRA-B2012-03_D13wk08’.

The test case can be demonstrated to run with one LTE UE (see section 5). Execution log is provided as evidence.

3. Verification Test Summary

Test Case:
11.2
Test Group:
IMS Notification
ATS Version:
iwd-EUTRA-B2012-03_D13wk08
System Simulator used:
Rohde & Schwarz Multi-RAT Protocol Tester
UE used:
XMM7160 Intel chip
Verification Status:
PASS
4. Corrections required
This section describes the TTCN changes required to ‘iwd-EUTRA-B2012-03_D13wk08’ release.
Change 1 - f_TC_11_2_IMS
	Function name
	 f_TC_11_2_IMS

	Reason for change
	1. Call to f_IMS_PTC_ImsInfo_GetURI() comes too early: URI not available yet at this point of time
2. UE waits for 30 secs and then comes back with next Register. We need to give it a bit more than 30 secs then rather than a bit less
3. According to 34.229-1, 11.2.4, step 4, we need to provide new SPIs and a new client protected port
4. Condition 2 needs to be passed to subsequent Register, see Change 2

	Summary of change
	1. Postpone call
2. Give 10% more time
3. Provide additional parameters to f_IMS_Register_SecurityInit()
4. Pass new parameter as document in Change 2

	TTCN module
	 IMS_34229/11/IMS_CC_NotificationTestcases

	MCC160 Comment
	


Before change

function f_TC_11_2_IMS() runs on IMS_PTC
  {

    var SecurityClientParams_Type v_SecurityClientParams;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var REGISTER_Request v_RegisterReq;

    var SUBSCRIBE_Request v_SubscribeReq;

    var charstring v_NW_AddrStr;

    var SipUrl v_ContactUrl;

    var template (value) Reginfo_Type v_XmlMessage;

    var template (value) RegInfo_RegistrationList_Type v_Registrations;

    var charstring v_PublicUserIdentity := f_IMS_PTC_ImsInfo_GetURI(firstId);
    var boolean v_ApplyGRUU := false;

    var integer v_NonceCount;

    timer t_TRegExpire;

    f_IMS_CC_Preamble(IPCAN_SignallingOnly, IMS_NULL);

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    v_RegisterReq := f_IMS_Register_NonGIBA();

    v_SubscribeReq := f_IMS_Register_SubscribeNotify(v_RegisterReq);

    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_RegisterReq.msgHeader);

    f_IMS_TestBody_Set(true);

    v_NW_AddrStr := f_IMS_PTC_NW_Address_GetHostFormat();

    // Set the 60 seconds expiration time in the XML body

    v_Registrations[0] := f_XML_RegistrationInfo(v_RegisterReq, v_PublicUserIdentity, "a100", "980", active, shortened, 60, v_ApplyGRUU);

    v_XmlMessage := f_XML_RegInfo(partial, 1, v_Registrations);

    f_IMS_Register_Notify(v_ContactUrl, v_SubscribeReq, v_XmlMessage, -, 2);

    f_IMS_CC_ReleaseConnection(IPCAN_InitialRegistration);

    // Step 3

    t_TRegExpire.start(30.0 * 0.9); // Wait for 30 seconds (-10%) for next registration

    f_IMS_CC_StartSignalling(IPCAN_MO_IMS_Signalling);

    v_NonceCount := 2;

    v_IMS_DATA_REQ := f_IMS_PTC_ReRegistration_WaitForRegistration(t_TRegExpire, v_NonceCount);

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    v_SecurityClientParams := f_IMS_PTC_ReRegistration_SecurityClientParamsGet();

    f_IMS_PTC_ReRegistration_SecurityClientParamsCheck(v_RegisterReq, v_SecurityClientParams);

    // Step 4

    f_IMS_Register_SecurityInit(v_RegisterReq, not4b valueof(tsc_IMS_AuthRAND));    /* generate a new RAND (NOTE: as workaround tsc_IMS_AuthRAND is a template => we need valueof) */

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));

    // Step 5

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest();
    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    // Step 6

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterReq))));

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "test case 11.2");

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MO_IMS_Signalling);

  }
After change

function f_TC_11_2_IMS() runs on IMS_PTC
  {

    var SecurityClientParams_Type v_SecurityClientParams;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var REGISTER_Request v_RegisterReq;

    var SUBSCRIBE_Request v_SubscribeReq;

    var charstring v_NW_AddrStr;

    var SipUrl v_ContactUrl;

    var template (value) Reginfo_Type v_XmlMessage;

    var template (value) RegInfo_RegistrationList_Type v_Registrations;

    var charstring v_PublicUserIdentity; 
    var boolean v_ApplyGRUU := false;

    var integer v_NonceCount;

    timer t_TRegExpire;

    f_IMS_CC_Preamble(IPCAN_SignallingOnly, IMS_NULL);

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    v_RegisterReq := f_IMS_Register_NonGIBA();

    v_SubscribeReq := f_IMS_Register_SubscribeNotify(v_RegisterReq);

    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_RegisterReq.msgHeader);

    v_PublicUserIdentity := f_IMS_PTC_ImsInfo_GetURI(firstId); 
    f_IMS_TestBody_Set(true);

    v_NW_AddrStr := f_IMS_PTC_NW_Address_GetHostFormat();

    // Set the 60 seconds expiration time in the XML body

    v_Registrations[0] := f_XML_RegistrationInfo(v_RegisterReq, v_PublicUserIdentity, "a100", "980", active, shortened, 60, v_ApplyGRUU);

    v_XmlMessage := f_XML_RegInfo(partial, 1, v_Registrations);

    f_IMS_Register_Notify(v_ContactUrl, v_SubscribeReq, v_XmlMessage, -, 2);

    f_IMS_CC_ReleaseConnection(IPCAN_InitialRegistration);

    // Step 3

    t_TRegExpire.start(30.0 * 1.1); // Wait for 30 seconds (+10%) for next registration 

    f_IMS_CC_StartSignalling(IPCAN_MO_IMS_Signalling);

    v_NonceCount := 2;

    v_IMS_DATA_REQ := f_IMS_PTC_ReRegistration_WaitForRegistration(t_TRegExpire, v_NonceCount);

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    v_SecurityClientParams := f_IMS_PTC_ReRegistration_SecurityClientParamsGet();

    f_IMS_PTC_ReRegistration_SecurityClientParamsCheck(v_RegisterReq, v_SecurityClientParams);

    // Step 4

    f_IMS_Register_SecurityInit(v_RegisterReq, not4b valueof(tsc_IMS_AuthRAND), -, tsc_IMS_Port_pc + 2, 258, 259); /* generate a new RAND (NOTE: as workaround tsc_IMS_AuthRAND is a template => we need valueof) */

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));

    // Step 5

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest(-, -, -, -, true); 
    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    // Step 6

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterReq))));

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "test case 11.2");

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MO_IMS_Signalling);

  }
Change 2 - f_IMS_REGISTER_SubsequentRequest
	Function name
	f_IMS_REGISTER_SubsequentRequest

	Reason for change
	Route header does not depend on NonceCount but on Condition A2 being true or not

	Summary of change
	Introduce parameter for A2 (see change 3) and use it


	TTCN module
	CommonIMS/IMS_CommonProcedures_Registration

	MCC160 Comment
	


Before change

function f_IMS_REGISTER_SubsequentRequest(template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit,
                                            boolean p_CondA7 := false,

                                            integer p_NonceCount := 1,

                                            template (omit) charstring p_ExplicitExpiry := omit) runs on IMS_PTC  return IMS_DATA_REQ

  { /* Receive subsequent REGISTER (condition A2) acc. to 34.229-1 Annex A.1.1

     * p_CondA7  .. condition A7 acc. to 34.229-1 Annex A.1.1 (Initial IMS emergency registration)

     * p_NonceCount  ..  Acc. to A.1.1 "counter to indicate how many times UE has sent the same value of nonce within successive REGISTERs, initial value shall be 1"

     *                   => during initial registration "1" is used per default; if p_NonceCount > 1 => Re-REGISTER

     */

    var IP_AddrInfo_Type v_UE_Address := f_IMS_PTC_UE_Address_Get();

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var REGISTER_Request v_RegisterReq;

    var MessageHeader v_MessageHeader;

    var InternetProtocol_Type v_Protocol;

    var boolean v_Match := false;

    // Receive/Get second REGISTER on secure port:

    v_IMS_DATA_REQ := fl_IMS_ReceiveOrGetRegisterRequest(p_IMS_DATA_REQ);

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    v_Protocol := v_IMS_DATA_REQ.RoutingInfo.Protocol;

    v_MessageHeader := v_RegisterReq.msgHeader;

    v_Match := match(v_MessageHeader, f_IMS_RegisterRequest_MessageHeaderRX_Subsequent(p_NonceCount));

    if (not v_Match) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid REGISTER Request");

    }

    else {

      f_IMS_A_1_1_MessageHeader_CommonChecks(v_MessageHeader, v_Protocol, p_CondA7, v_UE_Address, p_ExplicitExpiry); // @sic R5s120945 sic@

      if (f_IMS_PTC_ImsInfo_SigCompStarted()) {

        f_IMS_MessageHeader_CheckSigComp(v_MessageHeader);

      }

    }

    f_IMS_PTC_ImsInfo_SetContactUrl(f_MessageHeader_GetContactSipUrl(v_MessageHeader));

    return v_IMS_DATA_REQ;

  }
After change

function f_IMS_REGISTER_SubsequentRequest(template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit,
                                            boolean p_CondA7 := false,

                                            integer p_NonceCount := 1,

                                            template (omit) charstring p_ExplicitExpiry := omit,

                                            boolean p_CondA2 := true 
                                            ) runs on IMS_PTC  return IMS_DATA_REQ

  { /* Receive subsequent REGISTER (condition A2) acc. to 34.229-1 Annex A.1.1

     * p_CondA7  .. condition A7 acc. to 34.229-1 Annex A.1.1 (Initial IMS emergency registration)

     * p_NonceCount  ..  Acc. to A.1.1 "counter to indicate how many times UE has sent the same value of nonce within successive REGISTERs, initial value shall be 1"

     *                   => during initial registration "1" is used per default; if p_NonceCount > 1 => Re-REGISTER

     */

    var IP_AddrInfo_Type v_UE_Address := f_IMS_PTC_UE_Address_Get();

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var REGISTER_Request v_RegisterReq;

    var MessageHeader v_MessageHeader;

    var InternetProtocol_Type v_Protocol;

    var boolean v_Match := false;

    // Receive/Get second REGISTER on secure port:

    v_IMS_DATA_REQ := fl_IMS_ReceiveOrGetRegisterRequest(p_IMS_DATA_REQ);

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    v_Protocol := v_IMS_DATA_REQ.RoutingInfo.Protocol;

    v_MessageHeader := v_RegisterReq.msgHeader;

    v_Match := match(v_MessageHeader, f_IMS_RegisterRequest_MessageHeaderRX_Subsequent(p_NonceCount, p_CondA2)); 
    if (not v_Match) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid REGISTER Request");

    }

    else {

      f_IMS_A_1_1_MessageHeader_CommonChecks(v_MessageHeader, v_Protocol, p_CondA7, v_UE_Address, p_ExplicitExpiry); // @sic R5s120945 sic@

      if (f_IMS_PTC_ImsInfo_SigCompStarted()) {

        f_IMS_MessageHeader_CheckSigComp(v_MessageHeader);

      }

    }

    f_IMS_PTC_ImsInfo_SetContactUrl(f_MessageHeader_GetContactSipUrl(v_MessageHeader));

    return v_IMS_DATA_REQ;

  }
Change 3 - f_IMS_RegisterRequest_MessageHeaderRX_Subsequent
	Function name
	 f_IMS_RegisterRequest_MessageHeaderRX_Subsequent 

	Reason for change
	1. Route header does not depend on NonceCount but on Condition A2 being true or not.
2. To allow rport in  subsequent register

	Summary of change
	1. Introduce parameter for A2 and use it for Route header check
2. See R5s130187 for definition of cr_ViaDef2 (resp the alternative implementation in form of new function f_ViaParamsRX proposed by MCC160 in their comments to R5s130187, Change 6)

	TTCN module
	CommonIMS/IMS_CommonProcedures_Registration

	MCC160 Comment
	


Before change

function f_IMS_RegisterRequest_MessageHeaderRX_Subsequent(integer p_NonceCount := 1) runs on IMS_PTC return template (present) MessageHeader
  { /* subsequent REGISTER (condition A2) acc. to 34.229-1 Annex A.1.1

     * p_NonceCount  ..  Acc. to A.1.1 "counter to indicate how many times UE has sent the same value of nonce within successive REGISTERs, initial value shall be 1"

     *                   => during initial registration "1" is used per default; if p_NonceCount > 1 => Re-REGISTER

     */

    var template (present) MessageHeader v_MessageHeader_Register;

    var template (present) Route v_Route;

    var template (present) SipUrl v_SipUrlPublicUserId;

    var charstring v_Nonce := f_IMS_PTC_Security_GetNonce();

    var charstring v_NonceCountStr := "0000000" & int2str(p_NonceCount);     // assuming p_NonceCount < 9

    var IMS_ProtectedPorts_Type v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

    var integer v_CseqValue := f_IMS_PTC_ImsInfo_CseqIncr(register);

    v_SipUrlPublicUserId := f_SIP_BuildSipUrl_RX(f_IMS_PTC_ImsInfo_GetURI(registrationId)); /* public user identity is checked against what the UE has sent in initial REGISTER

                                                                                               (in case the UE does not have an ISIM this is the temporary user id)

                                                                                               => we don't need to consider A7 (emergency registration) */

    v_MessageHeader_Register := fl_IMS_RegisterRequest_MessageHeaderRX_Base(false, v_CseqValue);

    if (p_NonceCount == 1) {     // first security protected REGISTER

      v_Route := cr_RouteRegDef(px_IMS_Pcscf, v_Protected.Port_ps);

    } else {                     // Re-REGISTER

      v_Route := cr_RouteReReg(px_IMS_Pcscf, v_Protected.Port_ps, px_IMS_Scscf);

    }

    v_MessageHeader_Register.route       := v_Route ifpresent;

    v_MessageHeader_Register.via         := cr_ViaDef(-, cr_HostPort(?, v_Protected.Port_us));   // @sic R5s120727 change 7 sic@

    v_MessageHeader_Register.fromField   := cr_FromWithTag(v_SipUrlPublicUserId);  // checked in f_IMS_A_1_1_MessageHeader_CheckPublicUserIdentity

    v_MessageHeader_Register.toField     := cr_ToDef(v_SipUrlPublicUserId);        // checked in f_IMS_A_1_1_MessageHeader_CheckPublicUserIdentity

    v_MessageHeader_Register.contact     := cr_Contact(cr_SipUrl_HostPort(?, v_Protected.Port_us));   // checked in f_IMS_A_1_1_MessageHeader_CommonChecks; @sic R5s120727 change 7 sic@

    v_MessageHeader_Register.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *); // cell id only if applicable

    v_MessageHeader_Register.securityVerify     := f_IMS_PTC_BuildSecurityVerifyHeader();

    v_MessageHeader_Register.authorization      := cr_AuthorizationDef(px_IMS_Private_UserId, px_IMS_HomeDomainName, v_Nonce, tsc_IMS_Opaque, v_NonceCountStr);

    return v_MessageHeader_Register;

  }
After change

  function f_IMS_RegisterRequest_MessageHeaderRX_Subsequent(integer p_NonceCount := 1, boolean p_CondA2 := true) runs on IMS_PTC return template (present) MessageHeader 
  { /* subsequent REGISTER (condition A2) acc. to 34.229-1 Annex A.1.1

     * p_NonceCount  ..  Acc. to A.1.1 "counter to indicate how many times UE has sent the same value of nonce within successive REGISTERs, initial value shall be 1"

     *                   => during initial registration "1" is used per default; if p_NonceCount > 1 => Re-REGISTER

     */

    var template (present) MessageHeader v_MessageHeader_Register;

    var template (present) Route v_Route;

    var template (present) SipUrl v_SipUrlPublicUserId;

    var charstring v_Nonce := f_IMS_PTC_Security_GetNonce();

    var charstring v_NonceCountStr := "0000000" & int2str(p_NonceCount);     // assuming p_NonceCount < 9

    var IMS_ProtectedPorts_Type v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

    var integer v_CseqValue := f_IMS_PTC_ImsInfo_CseqIncr(register);

    v_SipUrlPublicUserId := f_SIP_BuildSipUrl_RX(f_IMS_PTC_ImsInfo_GetURI(registrationId)); /* public user identity is checked against what the UE has sent in initial REGISTER

                                                                                               (in case the UE does not have an ISIM this is the temporary user id)

                                                                                               => we don't need to consider A7 (emergency registration) */

    v_MessageHeader_Register := fl_IMS_RegisterRequest_MessageHeaderRX_Base(false, v_CseqValue);

    if (not p_CondA2) {     // first security protected REGISTER
      v_Route := cr_RouteRegDef(px_IMS_Pcscf, v_Protected.Port_ps);

    } else {                     // Re-REGISTER

      v_Route := cr_RouteReReg(px_IMS_Pcscf, v_Protected.Port_ps, px_IMS_Scscf);

    }

    v_MessageHeader_Register.route       := v_Route ifpresent;

    v_MessageHeader_Register.via         := cr_ViaDef2(-, cr_HostPort(?, v_Protected.Port_us));   // @sic R5s120727 change 7 sic@ 
    v_MessageHeader_Register.fromField   := cr_FromWithTag(v_SipUrlPublicUserId);  // checked in f_IMS_A_1_1_MessageHeader_CheckPublicUserIdentity

    v_MessageHeader_Register.toField     := cr_ToDef(v_SipUrlPublicUserId);        // checked in f_IMS_A_1_1_MessageHeader_CheckPublicUserIdentity

    v_MessageHeader_Register.contact     := cr_Contact(cr_SipUrl_HostPort(?, v_Protected.Port_us));   // checked in f_IMS_A_1_1_MessageHeader_CommonChecks; @sic R5s120727 change 7 sic@

    v_MessageHeader_Register.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *); // cell id only if applicable

    v_MessageHeader_Register.securityVerify     := f_IMS_PTC_BuildSecurityVerifyHeader();

    v_MessageHeader_Register.authorization      := cr_AuthorizationDef(px_IMS_Private_UserId, px_IMS_HomeDomainName, v_Nonce, tsc_IMS_Opaque, v_NonceCountStr);

    return v_MessageHeader_Register;

  }
Change 4 -  f_IMS_Register_SecurityInit
	Function name
	 f_IMS_Register_SecurityInit

	Reason for change
	See Change 1, bullet 3

	Summary of change
	Extend function with new parameters and pass them to templates cs_IMS_ProtectedPorts and cs_IMS_SPIs

	TTCN module
	 CommonIMS/IMS_CommonProcedures_Registration

	MCC160 Comment
	


Before change

function f_IMS_Register_SecurityInit(REGISTER_Request p_RegisterReq,
                                       template (value) B128_Type p_Rand := tsc_IMS_AuthRAND,

                                       boolean p_IsGIBA := false) runs on IMS_PTC

  { /* @sic R5s120858 change 6..8 - MCC160 implementation: additional parameter p_IsGIBA to deal with GIBA sic@ */

    var Common_AuthenticationParams_Type v_AuthenticationParams := f_IMS_AuthenticationInit(p_Rand);

    var IP_AddrInfo_Type v_UE_Address := f_IMS_PTC_UE_Address_Get();

    var IP_AddrInfo_Type v_NW_Address := f_IMS_PTC_NW_Address_Get();

    var SipUrl v_ContactUrl := f_MessageHeader_GetContactSipUrl(p_RegisterReq.msgHeader);

    var IMS_PortsAndSecurityConfigCnf_Type v_PortsAndSecurity;

    var PortNumber_Type v_Port_us;

    var PortNumber_Type v_Port_uc;

    var IPsec_SPI_Type v_SPI_us;

    var IPsec_SPI_Type v_SPI_uc;

    if (p_IsGIBA) {

      f_IMS_PortsAndSecurityConfig(f_SIP_ContactUrl_GetPort(v_ContactUrl),         // may be omit in which case 5060 is chosen by the IP PTC

                                   v_NW_Address,

                                   v_UE_Address);

    }

    else {

      f_IMS_InstallIPsecKeys(v_AuthenticationParams.IK, v_AuthenticationParams.CK);

      v_Port_us := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "port-s"));

      v_Port_uc := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "port-c"));

      v_SPI_us  := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "spi-s"));

      v_SPI_uc  := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "spi-c"));

      /* configure IMS security; any existing security context will be released

       * NOTE: release of the security context is not done at the end of the deregistration since it is not sure that the deregistration really happens and

       *       when realeasing the security directly after sending the deregistration it may lead to timing issues */

      v_PortsAndSecurity := f_IMS_PortsAndSecurityConfig(f_SIP_ContactUrl_GetPort(v_ContactUrl),         // may be omit in which case 5060 is chosen by the IP PTC

                                                         v_NW_Address,

                                                         v_UE_Address,

                                                         cs_IMS_SecurityInfo(cs_IMS_ProtectedPorts(v_Port_us, v_Port_uc),    // provides Port_us, Port_uc, Port_ps, Port_pc

                                                                             cs_IMS_SPIs(v_SPI_us, v_SPI_uc),                // provides SPI_us, SPI_uc (SPI_ps, SPI_pc will be set by the IP PTC)

                                                                             f_IMS_PTC_Security_GetIntegrityAlgorithm(),

                                                                             f_IMS_PTC_Security_GetCipheringAlgorithm()));

      /* v_PortsAndSecurity is now fully configured i.e. there are valid values for all ports and SPIs */

      f_IMS_PTC_Security_Init(v_AuthenticationParams, v_PortsAndSecurity.ProtectedPorts, v_PortsAndSecurity.SPIs);

    }

  }
After change

function f_IMS_Register_SecurityInit(REGISTER_Request p_RegisterReq,
                                       template (value) B128_Type p_Rand := tsc_IMS_AuthRAND,

                                       boolean p_IsGIBA := false,

                                       PortNumber_Type p_Port_pc := tsc_IMS_Port_pc,  
                                       template (omit)IPsec_SPI_Type p_SPI_ps := omit,
                                       template (omit)IPsec_SPI_Type p_SPI_pc := omit)
                                       runs on IMS_PTC

  { /* @sic R5s120858 change 6..8 - MCC160 implementation: additional parameter p_IsGIBA to deal with GIBA sic@ */

    var Common_AuthenticationParams_Type v_AuthenticationParams := f_IMS_AuthenticationInit(p_Rand);

    var IP_AddrInfo_Type v_UE_Address := f_IMS_PTC_UE_Address_Get();

    var IP_AddrInfo_Type v_NW_Address := f_IMS_PTC_NW_Address_Get();

    var SipUrl v_ContactUrl := f_MessageHeader_GetContactSipUrl(p_RegisterReq.msgHeader);

    var IMS_PortsAndSecurityConfigCnf_Type v_PortsAndSecurity;

    var PortNumber_Type v_Port_us;

    var PortNumber_Type v_Port_uc;

    var IPsec_SPI_Type v_SPI_us;

    var IPsec_SPI_Type v_SPI_uc;

    if (p_IsGIBA) {

      f_IMS_PortsAndSecurityConfig(f_SIP_ContactUrl_GetPort(v_ContactUrl),         // may be omit in which case 5060 is chosen by the IP PTC

                                   v_NW_Address,

                                   v_UE_Address);

    }

    else {

      f_IMS_InstallIPsecKeys(v_AuthenticationParams.IK, v_AuthenticationParams.CK);

      v_Port_us := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "port-s"));

      v_Port_uc := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "port-c"));

      v_SPI_us  := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "spi-s"));

      v_SPI_uc  := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "spi-c"));

      /* configure IMS security; any existing security context will be released

       * NOTE: release of the security context is not done at the end of the deregistration since it is not sure that the deregistration really happens and

       *       when realeasing the security directly after sending the deregistration it may lead to timing issues */

      v_PortsAndSecurity := f_IMS_PortsAndSecurityConfig(f_SIP_ContactUrl_GetPort(v_ContactUrl),         // may be omit in which case 5060 is chosen by the IP PTC

                                                         v_NW_Address,

                                                         v_UE_Address,

                                                         cs_IMS_SecurityInfo(cs_IMS_ProtectedPorts(v_Port_us, v_Port_uc, - , p_Port_pc),    // provides Port_us, Port_uc, Port_ps, Port_pc

                                                                             cs_IMS_SPIs(v_SPI_us, v_SPI_uc, p_SPI_ps, p_SPI_pc),           // provides SPI_us, SPI_uc (SPI_ps, SPI_pc will be set by the IP PTC) 
                                                                             f_IMS_PTC_Security_GetIntegrityAlgorithm(),

                                                                             f_IMS_PTC_Security_GetCipheringAlgorithm()));

      /* v_PortsAndSecurity is now fully configured i.e. there are valid values for all ports and SPIs */

      f_IMS_PTC_Security_Init(v_AuthenticationParams, v_PortsAndSecurity.ProtectedPorts, v_PortsAndSecurity.SPIs);

    }

  }
Change 5 - cs_IMS_SPIs
	Function name
	cs_IMS_SPIs

	Reason for change
	See Changes 1 and 4

	Summary of change
	Extend parameter list of cs_IMS_SPIs

	TTCN module
	CommonIMS/IMS_IP_Config

	MCC160 Comment
	


Before change

template (value) IMS_SPIs_Type cs_IMS_SPIs(IPsec_SPI_Type p_SPI_us,
                                             IPsec_SPI_Type p_SPI_uc) :=

  {

    SPI_us := p_SPI_us,

    SPI_uc := p_SPI_uc,

    SPI_ps := omit,  // to be assigned by the IP_PTC

    SPI_pc := omit   // to be assigned by the IP_PTC

  };
After change

template (value) IMS_SPIs_Type cs_IMS_SPIs(IPsec_SPI_Type p_SPI_us,
                                             IPsec_SPI_Type p_SPI_uc,

                                            template (omit)IPsec_SPI_Type p_SPI_ps := omit, 
                                            template (omit)IPsec_SPI_Type p_SPI_pc := omit) := 
  {

    SPI_us := p_SPI_us,

    SPI_uc := p_SPI_uc,

    SPI_ps := p_SPI_ps,  // to be assigned by the IP_PTC 
    SPI_pc := p_SPI_pc   // to be assigned by the IP_PTC 
  };
5. Execution Log Files

5.1  XMM7160 Intel chip
XMM7160 Intel chip passed this test case on the Rohde & Schwarz Multi-RAT Protocol Tester. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_11_2_IntelXMM7160.log:
In the log files (in txt format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6. References

[1] R5s130267: Supporting information for agreement of IMS test case 11.2 over LTE with 34.229-3 test model. This archive comprises txt format execution log file.

